
Change 1 

	Function Name
	fl_DNS_ServerGetIpAddress()

	Reason for change
	DNS Query with ‘AAAA’ sent for ipv6. Hence, IPv6 address needs to be retrieved instead of ipv4 address

	Summary of change
	Retrieved the IPv6 address from address resolution table

	Source of change
	Common\IP_PTC\DNS_Functions.ttcn

	MCC160 comments
	


Before change:

	function fl_DNS_ServerGetIpAddress(DNS_AddressResolutionTable_Type p_AddressResolutionTable,

                                     charstring p_DomainName,

                                     O2_Type p_QType) return charstring

  { /* get entry from the address resolution table matching the given domain name;

       return either IPv4 or IPv6 address depending on the qtype if entry is found or "" else */

    var integer i;

    var charstring v_Pattern;

    var charstring v_MatchingDomainName;

    var charstring v_IPAddr := "";

    for (i := 0; i < lengthof(p_AddressResolutionTable); i := i + 1) {

      v_Pattern := "(" & p_AddressResolutionTable[i].DomainNamePattern & ")";

      v_MatchingDomainName := regexp(p_DomainName, v_Pattern, 0);

      if (lengthof(v_MatchingDomainName) >  0) {

        select (p_QType) {

          case (tsc_DNSTYPE_A)    { v_IPAddr := p_AddressResolutionTable[i].IPv4Address; }

          case (tsc_DNSTYPE_AAAA) { v_IPAddr := p_AddressResolutionTable[i].IPv4Address; }
        }

        break;

      }

    }

    return v_IPAddr;

  }


After change :

	function fl_DNS_ServerGetIpAddress(DNS_AddressResolutionTable_Type p_AddressResolutionTable,

                                     charstring p_DomainName,

                                     O2_Type p_QType) return charstring

  { /* get entry from the address resolution table matching the given domain name;

       return either IPv4 or IPv6 address depending on the qtype if entry is found or "" else */

    var integer i;

    var charstring v_Pattern;

    var charstring v_MatchingDomainName;

    var charstring v_IPAddr := "";

    for (i := 0; i < lengthof(p_AddressResolutionTable); i := i + 1) {

      v_Pattern := "(" & p_AddressResolutionTable[i].DomainNamePattern & ")";

      v_MatchingDomainName := regexp(p_DomainName, v_Pattern, 0);

      if (lengthof(v_MatchingDomainName) >  0) {

        select (p_QType) {

          case (tsc_DNSTYPE_A)    { v_IPAddr := p_AddressResolutionTable[i].IPv4Address; }

          case (tsc_DNSTYPE_AAAA) { v_IPAddr := p_AddressResolutionTable[i].IPv6Address; } //tmp NK
        }

        break;

      }

    }

    return v_IPAddr;

  }


Change 2 

	Function Name
	fl_DNS_Answer()

	Reason for change
	DNS Query ‘AAAA’ sent for ipv6 address. Incorrection function called for ‘IPv4’. Hence, cs_DNS_RDATAu_AAAA  and f_Convert_IPv6Addr2OctString called in case of ipv6 address

	Summary of change
	cs_DNS_RDATAu_AAAA  and f_Convert_IPv6Addr2OctString used for ipv6 address

	Source of change
	Common\IP_PTC\DNS_Functions.ttcn

	MCC160 comments
	


Before change:

	  function fl_DNS_Answer(DNS_Question p_DNS_Question,

                         DNS_AddressResolutionTable_Type p_AddressResolutionTable) return template (value) DNS_RRs

  { /* generate answer(s) to a single question; returns {} when the question is not supported */

    var template (value) DNS_RRs v_DNS_RRs := {};

    var UInt16_Type v_DataLength;

    var template (value) DNS_RDATAu v_RData;

    var charstring v_DomainName := fl_DomainName_Decode(p_DNS_Question.qname);

    var O2_Type v_QType := p_DNS_Question.qtype;

    var charstring v_IpAddress;

    select (v_QType) {

      case (tsc_DNSTYPE_A, tsc_DNSTYPE_AAAA) {

        v_IpAddress := fl_DNS_ServerGetIpAddress(p_AddressResolutionTable, v_DomainName, v_QType);

        if (v_QType == tsc_DNSTYPE_AAAA) {

          v_DataLength := 4;

          v_RData := cs_DNS_RDATAu_A(f_Convert_IPv4Addr2OctString(v_IpAddress));

        } else {

          v_DataLength := 16;

          v_RData := cs_DNS_RDATAu_AAAA(f_Convert_IPv6Addr2OctString(v_IpAddress));
        }

        v_DNS_RRs := { cs_DNS_RR(p_DNS_Question.qname, v_QType, v_DataLength, v_RData) };

      }

      case else {

        f_ErrorLog(__FILE__, __LINE__, "unsupported DNS question (skipped)");

      }

    }

    return v_DNS_RRs;

  } 


After change :

	function fl_DNS_Answer(DNS_Question p_DNS_Question,

                         DNS_AddressResolutionTable_Type p_AddressResolutionTable) return template (value) DNS_RRs

  { /* generate answer(s) to a single question; returns {} when the question is not supported */

    var template (value) DNS_RRs v_DNS_RRs := {};

    var UInt16_Type v_DataLength;

    var template (value) DNS_RDATAu v_RData;

    var charstring v_DomainName := fl_DomainName_Decode(p_DNS_Question.qname);

    var O2_Type v_QType := p_DNS_Question.qtype;

    var charstring v_IpAddress;

    select (v_QType) {

      case (tsc_DNSTYPE_A, tsc_DNSTYPE_AAAA) {

        v_IpAddress := fl_DNS_ServerGetIpAddress(p_AddressResolutionTable, v_DomainName, v_QType);

        if (v_QType == tsc_DNSTYPE_A) { 
          v_DataLength := 4;

          v_RData := cs_DNS_RDATAu_A(f_Convert_IPv4Addr2OctString(v_IpAddress));

        } else if (v_QType == tsc_DNSTYPE_AAAA){

          v_DataLength := 16;

          v_RData := cs_DNS_RDATAu_AAAA(f_Convert_IPv6Addr2OctString(v_IpAddress));

        }
        v_DNS_RRs := { cs_DNS_RR(p_DNS_Question.qname, v_QType, v_DataLength, v_RData) };

      }

      case else {

        f_ErrorLog(__FILE__, __LINE__, "unsupported DNS question (skipped)");

      }

    }

    return v_DNS_RRs;

  }


